TOV equations for structure of spherical star

In Misner-Thorne-Wheeler texbook we have the following equations

$$\frac{dp}{dr} = -\frac{(\rho + p)(m + 4\pi r^3 p)}{r(r - 2m)}$$
$$\frac{dm}{dr} = 4\pi r^2 \rho$$
$$\frac{d\phi}{dr} = \frac{m + 4\pi r^3 p}{r(r - 2m)}$$

Here $\rho = \rho_0 + \epsilon$ where $\rho_0 = n_{\rm b} m_{\rm b}$ is rest-mass density and ϵ internal energy. We assume polytropic equation of state (see eqs. (1.74) and (1.86) in [Baumgarte&Shapiro])

$$p = k\rho_0^\gamma,$$

i.e.

$$\rho_0 = (p/k)^{1/\gamma} \quad \text{and} \quad \rho = \rho_0 + \frac{p}{\gamma - 1}$$

We put $k = 1$ and consider only dimensionless problem (see [Baumgarte&Shapiro] for scaling of true quantities such as stellar radius or mass for $k \neq 1$).

We get only inner part of the metric

$$ds^2 = -e^{2\Phi(r)} dt^2 + \frac{dr^2}{1 - \frac{2m(r)}{r}} + r^2 d\Omega^2$$

Note that the code does not solve function $\Phi(r)$ completely, because in the first step we simply assume $\Phi(r = 0) = 0$ which is not OK, the right boundary condition is $\Phi(\infty) = 0$ or $e^{2\phi(r_s)} = 1 - 2m(r_s)/r_s$ at surface of the star.

The following code illustrates the principles of BVP solution.

It is simplified to be clear enough and thus it does not exploit the usually very high precision of the pseudospectral method

```
In[24]:= solveTOV[pCentral_Real , OptionsPattern[
      {γ → 2.7 , dim → 9 , init → {rMax → 0.1 , m[0] → 0.01 , Φ[0] → -0.01 , p[0] → 0.01}}]] := Block[
    {γEOS = OptionValue[γ], n = OptionValue[dim], initGuess = OptionValue[init],
     rT, sl, residualΦ , residualP , residualM , bdryConditions ,
     chebRoots , rPts , eqns , unknowns , unknowns0 , solution},

    rT[n_Integer , r_] := ChebyshevT[n, -1 + 2 r / rMax];
    chebRoots[n_Integer ] := Table[ Cos[ (2 k + 1) Pi / 2 / n], {k, n - 1, 0, -1}] // N;
    rPts  = {r →  rMax (# + 1) / 2} & /@ chebRoots[n + 1];

    sl  = {p → Sum[ p[k] × rT[k, r], {k, 0, n}],
       Φ → Sum[Φ[k] × rT[k, r], {k, 0, n}],
       m → Sum[m[k] × rT[k, r], {k, 0, n}] r ^ 3
     };
     eos  =  ρ →  HeavisideTheta[p] ( Abs[p] ^ (1 / γEOS) + p / (γEOS - 1));

    residualΦ  = Hold[ D[Φ, r] - (m + 4 Pi r ^ 3 p) / r / (r - 2 m) ]  /. eos /. sl // ReleaseHold ;
    residualP  = Hold[ D[p, r] + (ρ + p) D[Φ, r] ]  /. eos /. sl // ReleaseHold ;
    residualM  = Hold[ D[m, r] - 4 Pi r ^ 2 ρ ]  /. eos /. sl // ReleaseHold ;

    bdryConditions  = Join[
       {p} /. sl /. r → rMax ,
       {Exp[2 Φ] - (1 - 2 m / r)} /. sl /. r → rMax ,
       {p - pCentral} /. sl /. r → 0
     ];

    eqns  = {residualM /. rPts[[2 ;;]], residualΦ /. rPts , residualP /. rPts[[ ;; -2]]}~
        Join~bdryConditions // Flatten ;

    unknowns  = {rMax}~Join~Table[{Φ[k], p[k], m[k]}, {k, 0, n}] // Flatten ;
    unknowns0  = Transpose[{unknowns ,
        unknowns /. initGuess /. {rMax → 0.1 , m[0] → 0.01 , Φ[0] → -0.01 , p[0] → 0.01} /.
         {m[k_] → 0, p[k_] → 0, Φ[k_] → 0}}];

    solution  = FindRoot[ eqns , unknowns0 , MaxIterations  → 200];

    Join[ solution  , {M → (m /. sl /. r → rMax /. solution),
       p → (p /. sl /. solution), m → (m /. sl /. solution), Φ → (Φ /. sl /. solution)}]
  ]
```

In[31]:=

```
With[{pC = 1.0},
 tmp = solveTOV[pC, dim → 16, init → {rMax → 0.5, Φ[0] → -0.5}];
 Print[Take[tmp, 7]];
 GraphicsRow[{
   Plot[m /. tmp, {r, 0, rMax /. tmp},
    PlotLabel → m, PlotTheme → "Detailed", PlotLegends → None],
   Plot[p /. tmp, {r, 0, rMax /. tmp}, PlotLabel → p,
    PlotTheme → "Detailed", PlotLegends → None],
   Plot[Φ /. tmp, {r, 0, rMax /. tmp}, PlotLabel → Φ,
    PlotTheme → "Detailed", PlotLegends → None]}]
]
```

```
{rMax → 0.404862 , Φ[0] → -1.02557 , p[0] → 0.449201 ,
 m[0] → 4.52608 , Φ[1] → 0.50898 , p[1] → -0.564426 , m[1] → -2.50833 }
```

Out[31]=



In[32]:=

```
prevSol = {rMax → 0.1, m[0] → 0.001, Φ[0] → -0.001, p[0] → 0.001};
Table[{rMax, M} /. (solveTOV[pC, dim → 8, γ → 2.7, init → prevSol]),
  {pC, 10^Range[-6, +0.5, 0.125]}];
ListLinePlot[%, PlotTheme → "Detailed", PlotLegends → None, FrameLabel → {R, m}]
```

Out[34]=